

Die Korrektheit von Mergesort

Christoph Lüth

23. Januar 2007

Definition von Mergesort und Formalisierung der Korrektheit

Die Funktion Mergesort ist wie folgt definiert:

```
msort :: [Int] -> [Int]
msort xs
  | length xs <= 1 = xs
  | otherwise = merge (msort front) (msort back) where
    (front, back) = splitAt ((length xs) `div` 2) xs
    merge :: [Int] -> [Int] -> [Int]
    merge [] x = x
    merge y [] = y
    merge (x:xs) (y:ys)
      | x <= y      = x:(merge xs (y:ys))
      | otherwise = y:(merge (x:xs) ys)
```

Zu zeigen ist die *Korrektheit von Mergesort*: die Rückgabe von Mergesort soll *sortiert* sein, und eine *Permutation* der Eingabeliste (d.h. dieselben Elemente in einer möglicherweise anderen Reihenfolge enthalten).

Sortiertheit definieren wir durch eine rekursive Haskell-Funktion:

```
sorted :: [Int] -> Bool
sorted []      = True
sorted [x]     = True
sorted (x:xs) = x <= head xs && sorted xs
```

Die Permutation definieren wir über eine Hilfsfunktion, die alle Elemente in der Liste zählt:

```
count :: Int -> [Int] -> Int
count z []      = 0
count z (x:xs) = if z == x then 1 + count z xs else count z xs
```

```
perm :: [Int] -> [Int] -> Bool
```

$$\text{perm } X Y \Leftrightarrow \forall z. \text{count } z X = \text{count } z Y \quad (1)$$

Wie wir sehen ist `perm` keine ausführbare Haskell-Funktion, weil wir über *alle möglichen Werte* von z (d.h. über alle ganzen Zahlen) quantifiziert haben. Das macht aber nichts, wir wollen `perm` ja schließlich nicht ausführen, sondern nur etwas damit beweisen.

Die Korrektheit von Mergesort (oder jeder anderen Sortierfunktion) ist damit folgende Formel:

$$\text{perm } X (\text{msort } X) \wedge \text{sorted } (\text{msort } X) \quad (2)$$

Anmerkung zur Notation: Wie in der Logik üblich unterdrücken wir außen stehende Allquantoren, d.h. wir schreiben (1) nicht als $\forall X Y. (\text{perm } X Y \Leftrightarrow \forall z. \text{count } z X = \text{count } z Y)$. Der innen stehende Allquantor über z ist allerdings unerlässlich, schließlich bedeutet $\text{perm } X Y \Leftrightarrow \text{count } z X = \text{count } z Y$ etwas ganz anderes als (1).

Eigenschaften von Permutation und Sortiertheit

Zuerst zeigen wir einen nützlichen Hilfssatz (Lemma) über die Funktion `count`:

Lemma 1

$$\text{count } z (X ++ Y) = \text{count } z X + \text{count } z Y$$

Beweis: Der Beweis erfolgt durch Induktion über das erste Argument X , weil $++$ durch primitive Rekursion über dieses Argument definiert ist.

Induktionsbasis:

$$\begin{aligned} \text{count } z ([] ++ Y) &= \text{count } z Y \\ &= 0 + \text{count } z Y \\ &= \text{count } z [] + \text{count } z Y \end{aligned}$$

Induktionsschritt: Die Induktionsvoraussetzung ist $\text{count } z (X ++ Y) = \text{count } z X + \text{count } z Y$. Wir unterscheiden zwei Fälle:

1. $x = z$

$$\begin{aligned} \text{count } z ((x : X) ++ Y) &= \text{count } z (x : (X ++ Y)) && \text{Def. von } ++ \\ &= 1 + \text{count } z (X ++ Y) && \text{Annahme: } x = z \\ &= 1 + \text{count } z X + \text{count } z Y && \text{nach Ind.vor.} \\ &= \text{count } z (x : X) + \text{count } z Y && \text{Annahme: } x = z \end{aligned}$$

2. $x \neq z$

$$\begin{aligned} \text{count } z ((x : X) ++ Y) &= \text{count } z (x : (X ++ Y)) && \text{Def. von } ++ \\ &= \text{count } z X ++ Y && \text{Annahme: } x \neq z \\ &= \text{count } z X + \text{count } z Y && \text{nach Ind.vor.} \\ &= \text{count } z (x : X) + \text{count } z Y && \text{Annahme: } x \neq z \end{aligned}$$

□

Im folgenden benutzen wir folgende *Notation*: Wir schreiben $X \sim Y$ für `perm X Y`. Das ist insbesondere nützlich, weil `perm` eine transitive und reflexive Relation ist, wie das folgende Lemma zeigt:

Lemma 2 (i) Wenn $X \sim Y$ und $Y \sim Z$, dann $X \sim Z$

(ii) $X \sim X$

Beweis (i):

$$\begin{aligned} &X \sim Y, Y \sim Z \\ \Leftrightarrow &\forall z_1. \text{count } z_1 X = \text{count } z_1 Y, \forall z_2. \text{count } z_2 Y = \text{count } z_2 Z \\ \Leftrightarrow &\forall z. \text{count } z X = \text{count } z Y, \text{count } z Y = \text{count } z Z \\ \Leftrightarrow &\forall z. \text{count } z X = \text{count } z Z \\ \Leftrightarrow &X \sim Z \end{aligned}$$

Beweis (ii):

$$X \sim X \Leftrightarrow \text{count } x X = \text{count } x X$$

□

Lemma 2 erlaubt uns Beweise über $\sim$ durch Verkettungen von Umformungen zu formulieren: beispielsweise zeigen wir mit folgender Kette $X \sim Y$:

$$\begin{aligned} X &\sim X_1 \\ &= X_2 \\ &\sim X_3 \\ &= Y \end{aligned}$$

Ferner ist $\sim$ auch symmetrisch, i.e. wenn $X \sim Y$, dann $Y \sim X$, und damit eine *Äquivalenzrelation*, aber das brauchen wir hier nicht. Wichtig ist aber, dass $\sim$ eine Kongruenz bezüglich der Listenkonkatenation $++$ ist:

Lemma 3 Wenn $A \sim X$ und $B \sim Y$, dann $A ++ B \sim X ++ Y$

Beweis:

$$\begin{aligned}
A \sim X, B \sim Y &\Leftrightarrow \forall z_1. \text{count } z_1 A = \text{count } z_1 X, \forall z_2. \text{count } z_2 B = \text{count } z_2 Y \\
&\Leftrightarrow \forall z. \text{count } z A = \text{count } z X, \text{count } z B = \text{count } z Y \\
&\Rightarrow \forall z. \text{count } z A + \text{count } z B = \text{count } z X + \text{count } z Y \\
&\Leftrightarrow \forall z. \text{count } z (A ++ B) = \text{count } z (X ++ Y) && \text{Nach Lemma 1} \\
&\Leftrightarrow A ++ B \sim X ++ Y
\end{aligned}$$

□

Nützliche Sonderfälle (Korollare) von Lemma 3 sind $A = X$, i.e. wenn $Y \sim B$, dann $A ++ B \sim A ++ Y$ sowie insbesondere $A = [a]$, i.e. wenn $X \sim Y$ dann $a : X \sim a : Y$, und analog $B = Y$. Ein weiterer Sonderfall ist $A = [a]$:

$$B \sim Y \Rightarrow a : B \sim a : Y \quad (3)$$

Wir benötigen ferner folgende Variation von Lemma 3:

Lemma 4

$$X ++ Y \sim Y ++ X$$

Der Beweis erfolgt analog zu Lemma 3.

Wir beschließend diesen Abschnitt mit einem Lemma über die Funktion `splitAt`:

Lemma 5 Sei `splitAt` n $X = (Y, Z)$, dann ist $X = Y ++ Z$.

Dieses ist weniger eine Eigenschaft als eher ein Teil der *Spezifikation* von `splitAt`. Der Beweis erfolgt durch Induktion über n , und bleibt dem Leser als Übung überlassen.

Die Funktion merge

Genauso wie bei Mergesort die eigentliche Arbeit von der Funktion `merge` erledigt wird, steckt die eigentliche Beweisarbeit für die Korrektheit von Mergesort in zwei Lemmata über die `merge` Funktion. `merge` ist durch allgemeine Rekursion über zwei Parameter gleichzeitig definiert, also benutzen wir als Beweisprinzip die Fixpunktinduktion.

Um eine Eigenschaft P über `merge` zu zeigen, d.h. $P(\text{merge } X \ Y)$ für ein beliebige X, Y , müssen wir also folgendes zeigen:

- (i) Induktionsbasis: $P([], Y)$ und $P(X, [])$ für beliebige X, Y ;
- (ii) Induktionsschritt:
Wenn $P(x : X, Y)$ und $P(X, y : Y)$, dann $P(x : X, y : Y)$.

Lemma 6 (merge bewahrt Permutation)

$$\text{merge } a \ b \sim a ++ b$$

Induktionsbasis: `merge` $a \ [] = a \sim a = a ++ []$, und `merge` $[] \ b = b \sim b = [] ++ b$.

Induktionsschritt: Wir betrachten `merge` $(a : as) (b : bs)$ und unterscheiden zwei Fälle:

1. $a \leq b$, dann ist

$$\begin{aligned}
\text{merge } (a : as) (b : bs) &= a : (\text{merge } as (b : bs)) \\
&\sim a : (as ++ b : bs) && \text{nach I.v. und (3)} \\
&= a : as ++ b : bs && \text{nach Def. ++}
\end{aligned}$$

2. $a > b$, dann ist

$$\begin{aligned}
\text{merge } (a : as) (b : bs) &= b : (\text{merge } (a : as) bs) \\
&\sim b : (a : as ++ bs) && \text{nach I.v. und (3)} \\
&\sim b : (bs ++ a : as) && \text{nach Lemma 4 und (3)} \\
&\sim b : bs ++ a : as && \text{nach Def. ++} \\
&\sim a : as ++ b : bs && \text{nach Lemma 4}
\end{aligned}$$

□

Lemma 7 (merge bewahrt Sortiertheit)

$$\text{sorted } X, \text{sorted } Y \implies \text{sorted } (\text{merge } X Y)$$

Induktionsbasis:

- $\text{merge } X [] = X$, damit $\text{sorted } (\text{merge } X [])$ wenn $\text{sorted } X$.
- $\text{merge } [] Y = Y$, damit $\text{sorted } (\text{merge } [] Y)$ wenn $\text{sorted } Y$.

Induktionsschritt: Auch hier unterscheiden wir zwei Fälle:

1. $x \leq y$

$$\begin{aligned}
&\text{sorted } (\text{merge } (x : X) (y : Y)) \\
&\Leftrightarrow \text{sorted } (x : \text{merge } X (y : Y)) \\
&\Leftrightarrow x \leq \text{head } (\text{merge } X (y : Y)) \wedge \text{sorted } (\text{merge } X (y : Y)) \quad \text{nach Def. sorted}
\end{aligned}$$

Der zweite Teil der Konjunktion ist die Induktionsvoraussetzung, wir müssen also nur den ersten Teil zeigen. Es gilt folgende Disjunktion (da mindestens $X \neq []$ oder $Y \neq []$):

$$\text{head } (\text{merge } X Y) = \text{head } X \vee \text{head } (\text{merge } X Y) = \text{head } Y \quad (4)$$

Damit ist entweder $\text{head } (\text{merge } X (y : Y)) = y$, und nach Voraussetzung $x \leq y$; oder $\text{head } (\text{merge } X (y : Y)) = \text{head } X$, und dann ist $x < \text{head } X$, was wegen der Voraussetzung $\text{sorted } (x : X)$ gilt.

2. $x > y$, ist völlig analog:

$$\begin{aligned}
&\text{sorted } \text{merge } (x : X) (y : Y) \\
&\Leftrightarrow \text{sorted } (y : \text{merge } (x : X) Y) \\
&\Leftrightarrow y \leq \text{head } (\text{merge } (x : X) Y) \wedge \text{sorted } (\text{merge } (x : X) Y) \quad \text{nach Def.}
\end{aligned}$$

Der zweite Teil der Konjunktion ist die Induktionsvoraussetzung. Für den ersten Teil ist mit (4) entweder $\text{head } (\text{merge } (x : X) Y) = x$, und nach Voraussetzung $y < x$; oder $\text{head } (\text{merge } (x : X) Y) = \text{head } Y$, und dann ist $y < \text{head } Y$, was wegen der Voraussetzung $\text{sorted } (y : Y)$ gilt.

□

Die Korrektheit von Mergesort

Wir können jetzt die Korrektheit von Mergesort zeigen, i.e. Gleichung (2) in zwei Teilen:

(i)

$$\text{perm } (\text{msort } X) X,$$

(ii)

$$\text{sorted } (\text{msort } X).$$

Das Beweisprinzip hierbei muß der Definition von Mergesort entsprechen. Mergesort ist rekursiv über der *Länge* der Liste definiert; die Basis sind Listen der Länge eins oder null, und die im Rekursionsschritt wird die Länge der Liste halbiert. Deshalb zeigen wir die Behauptungen (i) und (ii) durch *Induktion über der Länge der Liste*¹. Aus Gründen der Übersichtlichkeit zeigen wir beide Teile getrennt.

Beweis (i):

Induktionsbasis: Wenn $\text{length } X \leq 1$, dann gilt durch Einsetzen der Funktionsdefinition $\text{msort } X = X \sim X$.

Induktionsschritt: Zu zeigen:

$$\text{merge } (\text{msort } \text{front}) (\text{msort } \text{back}) \sim X$$

mit $(\text{front}, \text{back}) = \text{splitAt } (\text{length } X / 2) X$.

Die Induktionsannahme ist

$$\text{front} \sim \text{msort } \text{front}, \text{back} \sim \text{msort } \text{back}.$$

Nach Lemma 5 ist $X = \text{front} ++ \text{back}$. Damit ist

$$\begin{aligned} X &= \text{front} ++ \text{back} \\ &\sim \text{msort } \text{front} ++ \text{msort } \text{back} && \text{nach Ind.vor. und Lemma 3} \\ &\sim \text{merge } (\text{msort } \text{front}) (\text{msort } \text{back}) && \text{nach Lemma 6} \end{aligned}$$

Beweis (ii):

Induktionsbasis: Wenn $\text{length } X \leq 1$, dann gilt durch Einsetzen der Funktionsdefinition $\text{msort } X = X$. Sowohl für $X = []$ als auch für $X = [x]$ folgt direkt $\text{sorted } X$, und damit $\text{sorted } (\text{msort } X)$.

Induktionsschritt: Zu zeigen ist

$$\text{sorted } (\text{merge } (\text{msort } \text{front}) (\text{msort } \text{back})).$$

Die Induktionssvoraussetzung ist

$$\text{sorted } (\text{msort } \text{front}), \text{sorted } (\text{msort } \text{back}).$$

Damit wird der Induktionsschritt direkt durch Lemma 7 bewiesen.

□

Zusammenfassung

Wir haben die Korrektheit eines nicht-trivialen Algorithmus spezifiziert und bewiesen. Die Grundidee des Beweises ist eine Rekursion, die genau dem rekursiven Aufbau der Funktion folgt.

Wir brauchten am Anfang etwas zusätzliches Rüstzeug (wie die Lemmas über Permutation, und die nützliche Schreibweise der Permutation). Das liegt daran, dass “Sortiertheit” eine nicht-triviale Eigenschaft ist. Auch das ist typisch: um solch eine nicht-triviale Eigenschaft zu zeigen, muss man erst verschiedene Hilfsannahmen über diese Eigenschaft zeigen. Es ist dabei oft hilfreich, diese Lemmata allgemeiner zu zeigen als sie auf den ersten Blick gebraucht werden, weil man sie vielleicht öfter braucht (vgl. unser Lemma 3). Außerdem ist bei induktiven Beweisen der Beweis einer allgemeineren Annahme oft einfacher, weil eine allgemeinere Annahme eine stärkere Induktionsvoraussetzung ist.

¹Dieses ist keine *natürliche Induktion*, sondern tatsächlich eine Fixpunktinduktion.